GPCT method: SAS implementation

Johns et al. Generalised pairwise comparisons for trend: An extension to the win ratio and win odds for dose-response and prognostic variable analysis with arbitrary statements of outcome preference. Statistical Methods in Medical Research 2023, Vol. 32(3) 609–625

1. Introduction

It is often the case in medical research that clinicians want to investigate either a dose–response relationship between some outcome of interest and the amount of received therapy or the relationship between the outcome and some well-ordered prognostic variable. While there are established statistical methods enabling such analyses of binary, nominal, ordinal and categorical outcomes, complex and multifaceted outcome measures are more challenging to investigate in this fashion. The win ratio is a recently proposed generalized pairwise comparison (GPC) method that addresses this challenge by testing for preference between two groups based on pairwise comparisons on a lexicographically ordered set of outcomes of interest. The GPCT (GPC for Trend) is an extension of the win ratio to the case of dose–response and prognostic variable analysis on outcomes with arbitrary statements of preference.

GPCT is defined as the odds that for a random pair of observation (X_1, Y_1) and (X_2, Y_2), if $X_1 > X_2$ then $Y_1 \succ Y_2$, which means "Y_1 is preferred to Y_2", and may be understood as a generalization of inequality tests.

2. GPCT method

2.1 The GPCT statistic

The GPCT statistic is calculated as follows. Let $Obs_i = (x_i, y_i, p_i)$ be the ith observation , where x_i is the observed value of the explanatory variable, y_i is the observed (possibly complex and multifaceted) outcome, and p_i is a weight associated with the observation. The introduction of weights is used to link GPCT to GenOR, which is evaluated on cross-tabulated ordinal data. A single observation has weight 1, and if multiple, identical sets of (x_i, y_i) were observed, they may be replaced with a single observation with weight set to the sum count of observed instances.

First, define two functions $K(y_i, y_j)$ and $L(x_i, x_j)$ for comparing the ith and jth observations.

$$K(y_i, y_j) = \begin{cases} 1, & if\ y_i \succ y_j \\ -1, & if\ y_i \prec y_j \\ 0, & if\ y_i = y_j \end{cases}$$

$$L(x_i, x_j) = \begin{cases} 1, & if\ x_i > x_j \\ -1, & if\ x_i < x_j \\ undefined, & if\ x_i = x_j \end{cases}$$

Then define Rs_i, Rd_i and Rt_i as the probability that, in comparing a patient (x_i, y_i) to a random patient (x_j, y_j), the pair will be concordant, discordant, or tied, respectively. The definitions of concordant and discordant are analogous to those used in Agresti's generalised odds ratio on a cross-tabulated summary of two ordinal variables, though in this form may be evaluated on any type of data. They are calculated as follows:

$$Rs_i = P(Patient\ i\ is\ concordant\ with\ a\ randomly\ drawn\ observation)$$
$$= \frac{1}{\sum_{j=1}^{N} p_j} \sum_{j=1}^{N} \begin{cases} p_j, if\ K(y_i, y_j) \times L(x_i, x_j) = 1 \\ 0, otherwise \end{cases}$$

$$Rd_i = P(Patient\ i\ is\ discordant\ with\ a\ randomly\ drawn\ observation)$$

FIRMA CLINICAL RESEARCH

$$= \frac{1}{\sum_{j=1}^{N} p_j} \sum_{j=1}^{N} \begin{cases} p_j, if\ K(y_i, y_j) \times L(x_i, x_j) = -1 \\ 0, otherwise \end{cases}$$

$$Rt_i = P(Patient\ i\ is\ tied\ with\ a\ randomly\ drawn\ observation)$$
$$= \frac{1}{\sum_{j=1}^{N} p_j} \sum_{j=1}^{N} \begin{cases} p_j, if\ K(y_i, y_j) \times L(x_i, x_j) = 0 \\ 0, otherwise \end{cases}$$

At this point, we may either split ties evenly between being considered concordant and discordant to calculate the trend odds, or drop them to calculate the trend ratio. If trend odds is to be calculated, we reflect this by redefining Rs_i and Rd_i as

$$Rs_i^* = Rs_i + \frac{1}{2} Rt_i$$

$$Rd_i^* = Rd_i + \frac{1}{2} Rt_i$$

If the trend ratio is to be calculated, we omit this step by simply setting $Rs_i^* = Rs_i$ and $Rd_i^* = Rd_i$.

With the probability of concordance and discordance for each observation calculated, we can now calculate the probability of a random pair of observations will be concordant or discordant. The probability of concordance and discordance is calculated by summing across each patient, and are given by:

$$P_c = P(A\ random\ pair\ of\ observation\ is\ concordant)$$
$$= \frac{1}{\sum_{i=1}^{N} p_i} \sum_{i=1}^{N} p_i Rs_i^*$$

$$P_d = P(A\ random\ pair\ of\ observation\ is\ discordant)$$
$$= \frac{1}{\sum_{i=1}^{N} p_i} \sum_{i=1}^{N} p_i Rd_i^*$$

The GPCT statistic, which is either the trend ratio or trend odds depending on how tied pairs are treated, is then defined as

$$GPCT = \frac{P_c}{P_d}$$

with SE given by the quantity

$$SE(GPCT) = \frac{2}{P_d} \left[\sum_{i=1}^{N} \frac{p_i}{\sum_{i=1}^{N} p_i} (GPCT \times Rd_i^* - Rs_i^*)^2 \Big/ \sum_{i=1}^{N} p_i \right]^{1/2}$$

2.2 Statistical Inference

To construct confidence interval and perform hypothesis testing, we should first know that $\ln(GPCT)$ is normally distributed with its standard error calculated using the delta method and given by

$$SE(\ln(GPCT)) = \frac{SE(GPCT)}{GPCT}$$

It is important to note that the GPCT statistic tests for trend in outcome preference as a function of some explanatory variable. That is, the null hypothesis of this test for either the trend odds or trend ratio is given by

$$H_0: GPCT = 1$$
$$H_{A(2-sided)}: GPCT \neq 1$$

while the null hypothesis is true when there is no relationship between an explanatory variable and patient outcomes, it may also be true in the presence of non-monotonic relationships. In clinical terms, the null would be interpreted 'a higher dose is not associated with better outcomes', or 'greater injury severity is not associated with worse outcomes'. For example, consider a trial of three escalating treatment doses (with equal treatment allocation) that achieve a binary positive outcome with probabilities of 0.2, 0.7, and 0.2, respectively, where the middle group is the optimal dose and the higher dose is toxic. In this situation, the null hypothesis would still be correct, as the odds that a random observation with a larger dose would achieve a better outcome would be 1. While this behavior is common to measures of trend (e.g. logistic regression would give similar results in this case), care must be taken in the interpretation of the trend statistic

For hypothesis testing above, p-value and $(1-\alpha)\times 100\%$ confidence interval could be provided by:

$$p-value = 2\times\left(1-\Phi\left(\left|\frac{\ln(GPCT)}{SE(\ln(GPCT))}\right|\right)\right) = 2\times\left(1-\Phi\left(\left|\frac{GPCT\times\ln(GPCT)}{SE(GPCT)}\right|\right)\right)$$

$$(1-\alpha)\times 100\%\ CI = \left[exp\left(\ln(GPCT)\pm Z_{1-\alpha/2}\times SE(\ln(GPCT))\right)\right] = \left[exp\left(\ln(GPCT)\pm Z_{1-\alpha/2}\times\frac{SE(GPCT)}{GPCT}\right)\right]$$

Besides, we can perform an overall hypothesis testing for stratified GPCT. Let $GPCT_m$ be the value of GPCT belonging to the mth stratum, where $m = 1, 2, \ldots, M$. A pooled estimate $GPCT_{pool}$ across these M strata is given by

$$GPCT_{pool} = \exp\left[\sum_{m=1}^{M}\frac{\ln(GPCT_m)}{\mathrm{SE}(\ln(GPCT_m))^2}\Big/\sum_{m=1}^{M}\frac{1}{\mathrm{SE}(\ln(GPCT_m))^2}\right]$$

Which is log-normally distributed with standard error given by

$$\mathrm{SE}\big(\ln\big(GPCT_{pool}\big)\big) = \sqrt{\left(\sum_{m=1}^{M}\frac{1}{\mathrm{SE}(\ln(GPCT_m))^2}\right)^{-1}}$$

This pooled estimate may also be used to test the hypothesis $H_0: GPCT_1 = GPCT_2 = \cdots = GPCT_M$ by using it to calculate the test statistic

$$V = \sum_{m=1}^{M}\frac{\big(\ln(GPCT_m)-\ln\big(GPCT_{pool}\big)\big)^2}{\mathrm{SE}(\ln(GPCT_m))^2}$$

Which is χ^2 distributed with $M-1$ degrees of freedom.

3. SAS implementation

To implement GPCT method in SAS, a macro has been developed with five required and one optional parameters (**_STRARUM**):

%macro GPCT(_DATSRC =, _STRATUM =, _X =, _Y =, _TREND =, _ALPHA =);

_DATSRC refers to the input dataset containing the necessary data for analysis.
_STRATUM refers to the space-separated variable name of the stratum, which can be any variable containing stratum information that the user wishes to detect trends within. If not specified, the whole **_X** and **_Y** would be considered of one stratum.
_X refers to the variable name of the explanatory variable, whose values could be compared numerically.
_Y refers to the variable name of the observed outcome, whose values could be compared numerically. It is assumed that higher values indicate better outcomes.
_TREND takes a string value of either "**odds**" and "**ratio**" to indicate whether it is trend odds or trend ratio to be calculated.
_ALPHA takes a value between 0 and 1 (0.05 by default), representing the desired level of significance for creating the confidence interval. Specifically, it determines the $(1-\alpha)\times 100\%$ confidence interval.

Two datasets will be generated:

GPCT with $P_c, P_d, GPCT, SE(GPCT), (1-\alpha)\ CI$ and p-value for each stratum group $H_0: GPCT = 1$.

GPCT_OVERALL_TEST with V, DF (degrees of freedom) and p-value for the overall hypothesis testing $H_0: GPCT_1 = GPCT_2 = \cdots = GPCT_m$.

%macro GPCT is proposed here:

```
%macro GPCT(_DATSRC =, _STRATUM =, _X =, _Y =, _TREND = , _ALPHA = 0.05);

%if %bquote(%upcase(&_TREND.)) eq %str(ODDS) %then %do;
%end;
%else %if %bquote(%upcase(&_TREND.)) eq %str(RATIO) %then %do;
%end;
%else %do;
    %put %str(WAR)NING: _TREND takes value as odds or ratio. Macro stop.;
    %return;
%end;

%if %bquote(&_STRATUM.) eq %str() %then %do;
    %let VAR_KEEP = _STRATUM &_X. &_Y.;
%end;
%else %do;
    %let VAR_KEEP = &_STRATUM. &_X. &_Y.;
%end;

%macro SQLCond(_VarList =, _deli = %str( ), _left = , _right =, _out =);
%local COND i;
%do i = 1 %to %sysfunc(countw(&_VarList., "&_deli."));

    %let ELEMENT = %upcase(%scan(&_VarList., &i, "&_deli."));
```

```
    %if %length(&COND) %then %do;
        %let COND = &COND and &_left..&ELEMENT. = &_right..&ELEMENT.;
    %end;
    %else %do;
        %let COND = &_left..&ELEMENT. =  &_right..&ELEMENT.;
    %end;

%end;
%let &_out = &COND;
%mend SQLCond;

%macro SQLSelect(_VarList =, _deli = %str( ), _prefix =, _out =);
%local COND i;
%do i = 1 %to %sysfunc(countw(&_VarList., "&_deli."));

    %let ELEMENT = %upcase(%scan(&_VarList., &i, "&_deli."));
    %if %length(&COND) %then %do;
        %let COND = &COND, &_prefix..&ELEMENT.;
    %end;
    %else %do;
        %let COND = &_prefix..&ELEMENT.;
    %end;
%end;
%let &_out = &COND;
%mend SQLSelect;


data _GPCT_data;
    set &_DATSRC.;
%if %bquote(&_STRATUM.) eq %str() %then %do;
    _STRATUM = 1;
%end;
run;


%if %bquote(&_STRATUM.) eq %str() %then %do;
    %let _STRATUM = _STRATUM;
%end;

%let VAR_KEEP_SQL = %sysfunc(tranwrd(&VAR_KEEP., %str( ), %str(, )));
%let _STRATUM_SQL = %sysfunc(tranwrd(&_STRATUM., %str( ), %str(, )));
%local _STRATUM_SQL_a   ; %SQLSelect(_VarList = &_STRATUM., _deli = %str(
), _prefix =a           , _out = _STRATUM_SQL_a   );
%local _STRATUM_SQL_b   ; %SQLSelect(_VarList = &_STRATUM., _deli = %str(
), _prefix =b           , _out = _STRATUM_SQL_b   );
%local _STRATUM_SQL_COND; %SQLCond  (_VarList = &_STRATUM., _deli = %str(
), _left = a, _right = b, _out = _STRATUM_SQL_COND);
%put &VAR_KEEP.;
%put &VAR_KEEP_SQL.;
%put &_STRATUM.;
```

```
%put &_STRATUM_SQL.;
%put &_STRATUM_SQL_a.;
%put &_STRATUM_SQL_b.;
%put &_STRATUM_SQL_COND.;



*** Determine weight associated with each observation ***;

proc sql noprint;
    create table _GPCT01 as
    select *, count(*) as _P, monotonic() as ID
    from _GPCT_data(keep = &VAR_KEEP.)
    group by &VAR_KEEP_SQL.
    ;
quit;

*** Compute ∑_{i=1}^{N} p_i of each treatment group to be used later ***;

proc sql noprint;
    create table _P_SUM as
    select &_STRATUM_SQL., sum(_P) as P_SUM
    from _GPCT01
    group by &_STRATUM_SQL.
    ;
quit;

*** List all patient pair within the same treatment group ***;


proc sql noprint;
    create table _GPCT02 as
    select &_STRATUM_SQL_a., a.&_X. as Xi, a.&_Y. as Yi, a._P as Pi,
b.&_X. as Xj, b.&_Y. as Yj, b._p as Pj
    from _GPCT01 a, _GPCT01 b
    where &_STRATUM_SQL_COND.
    ;
quit;

*** Apply function K and L on each patient pair to be used for Rs_i, Rd_i, Rt_i
later  ***;


data _GPCT03;
    set _GPCT02;
    K = ifn(Yi < Yj, -1, ifn(Yi > Yj, 1, 0));
    L = ifn(Xi < Xj, -1, ifn(Xi > Xj, 1, .));
    KL = K * L;
    _Rsi = ifn(KL =  1, Pj, 0);
    _Rdi = ifn(KL = -1, Pj, 0);
```

```
    _Rti = ifn(KL =  0, Pj, 0);
run;

proc sql noprint;
    create table _GPCT04 as
    select &_STRATUM_SQL., Xi, Yi, Pi, sum(_Rsi) as _Rsi_SUM, sum(_Rdi) as
_Rdi_SUM, sum(_Rti) as _Rti_SUM
    from _GPCT03
    group by &_STRATUM_SQL., Xi, Yi, Pi
    ;
quit;

proc sql noprint;
    create table _GPCT05 as
    select a.*, b.P_SUM
    from _GPCT04 a left join _P_SUM b
    on &_STRATUM_SQL_COND.
    ;
quit;

*** Compute Rs_i,Rd_i,Rt_i ***;


proc sql noprint;
    create table _GPCT06 as
    select &_STRATUM_SQL., Xi, Yi, Pi, P_SUM, _Rsi_SUM / P_SUM as Rsi_PRE,
_Rdi_SUM / P_SUM as Rdi_PRE, _Rti_SUM / P_SUM as Rti
    from _GPCT05
    ;
quit;

*** Compute Rs_i^*,Rd_i^* ***;


proc sql noprint;
    create table _GPCT07 as
    %if %upcase(&_TREND.) = ODDS %then %do;
    select &_STRATUM_SQL., Xi, Yi, Pi, P_SUM, Rsi_PRE + Rti / 2 as Rsi,
Rdi_PRE + Rti / 2 as Rdi, Rti
    %end;
    %else %if %upcase(&_TREND.) = RATIO %then %do;
    select &_STRATUM_SQL., Xi, Yi, Pi, P_SUM, Rsi_PRE as Rsi, Rdi_PRE as
Rdi, Rti
    %end;
    from _GPCT06 a
    ;
quit;

*** Compute GPCT statistic ***;

*** Point Estimation ***;
```

```
proc sql noprint;
    create table _GPCT08_PE as
    select &_STRATUM_SQL., sum(Rsi * Pi) / P_SUM as Pc, sum(Rdi * Pi) /
P_SUM as Pd, calculated Pc / calculated Pd as GPCT
    from _GPCT07
    group by &_STRATUM_SQL., P_SUM
    ;
quit;
*** done ***;

*** Compute SE(GPCT) ***;

proc sql noprint;
    create table _GPCT09 as
    select a.*, b.Pc, b.Pd, b.GPCT
    from _GPCT07 a left join _GPCT08_PE b
    on &_STRATUM_SQL_COND.
    ;
quit;

proc sql noprint;
    create table _GPCT10 as
    select &_STRATUM_SQL., sqrt(sum((GPCT * Rdi - Rsi)**2 * Pi / P_SUM) /
P_SUM) * 2 / Pd as GPCT_SE
    from _GPCT09
    group by &_STRATUM_SQL., Pd, P_SUM
    ;
quit;

proc sql noprint;
    create table _GPCT11_SE as
    select a.*, b.GPCT_SE
    from _GPCT08_PE a left join _GPCT10 b
    on &_STRATUM_SQL_COND.
    ;
quit;
*** done ***;

*** test Ho: GPCT = 1 ***;
    *** ln(GPCT) is normally distributed with SE(ln(GPCT)) = SE(GPCT) /
GPCT ***;
     *** Perform hypothesis testing compute p-value and construct (1-
alpha)% CI ***;

    *** equivalent to test H0: ln(GPCT) = 0 ***;
data _GPCT12;
    set _GPCT11_SE;
    ln_GPCT = log(GPCT);
    ln_GPCT_SE = GPCT_SE / GPCT;
    ln_GPCT_LCI = ln_GPCT + (probit((&_ALPHA./2)) * ln_GPCT_SE);
    ln_GPCT_UCI = ln_GPCT + (probit(1-(&_ALPHA./2)) * ln_GPCT_SE);
```

```
    GPCT_LCI = exp(ln_GPCT_LCI);
    GPCT_UCI = exp(ln_GPCT_UCI);
    PVALUE = 2 * (1 - cdf("NORMAL", abs(ln_GPCT / ln_GPCT_SE)));
run;
*** done ***;

*** test overall stratum Ho: GPCT_1 = GPCT_2 = … = GPCT_M ***;
data _GPCT_STR01;
    set _GPCT12;
    GPCT_STR_NUME_PRE = ln_GPCT / (ln_GPCT_SE)**2;
    GPCT_STR_DENO_PRE = 1 / (ln_GPCT_SE)**2;
run;

proc sql noprint;
    create table _GPCT_STR02 as
    select *, sum(GPCT_STR_NUME_PRE) as NUME_PRE, sum(GPCT_STR_DENO_PRE)
as DENO_PRE
    from _GPCT_STR01
    ;
quit;

data _GPCT_STR03;
    set _GPCT_STR02;
    ln_GPCT_POOL = NUME_PRE / DENO_PRE;
run;

data _GPCT_STR04;
    set _GPCT_STR03;
    V_PRE = (ln_GPCT - ln_GPCT_POOL)**2 / ln_GPCT_SE**2;
run;

proc sql noprint;
    create table _GPCT_STR05 as
    select sum(V_PRE) as V, count(*) - 1 as DF, (1 - cdf("CHISQUARE",
calculated V, calculated DF)) as PVALUE
    from _GPCT_STR04
    ;
quit;

*** done ***;

data GPCT;
    set _GPCT12;
%if %bquote(&_STRATUM.) eq _STRATUM %then %do;
    keep Pc Pd GPCT GPCT_SE GPCT_LCI GPCT_UCI PVALUE;
%end;
%else %do;
    keep &_STRATUM. Pc Pd GPCT GPCT_SE GPCT_LCI GPCT_UCI PVALUE;
%end;
run;
```

```
data GPCT_OVERALL_TEST;
    set _GPCT_STR05;
run;

proc datasets library = work memtype = data nolist;
    delete _GPCT: _P_SUM;
run; quit;
*** done ***;

%mend GPCT;
```

```
%let seed = 156168;
%let split = 1000;
data dummy;
    length CATE $17. TRT $30.;
    CATE = "Varying Slope";

    do i = 0 to &split.;
        SUBJID = "01-"||put(i, z4.);
        TRT = "Y ~ N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = NOISE;
    output;
    end;

    do i = 0 to &split.;
        SUBJID = "02-"||put(i, z4.);
        TRT = "Y ~ X + N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = X + NOISE;
    output;
    end;

    do i = 0 to &split.;
        SUBJID = "03-"||put(i, z4.);
        TRT = "Y ~ 2X + N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = 2 * X + NOISE;
    output;
    end;

    do i = 0 to &split.;
        SUBJID = "04-"||put(i, z4.);
        TRT = "Y ~ 10X * (1 - X) + N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = 10 * X * (1 - X) + NOISE;
    output;
    end;

    CATE = "Varying Intercept";
    do i = 0 to &split.;
        SUBJID = "05-"||put(i, z4.);
        TRT = "Y ~ X + 1 + N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = X + 1 + NOISE;
```

```
    output;
    end;

    do i = 0 to &split.;
        SUBJID = "06-"||put(i, z4.);
        TRT = "Y ~ X + 2 + N(0, 0.5)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(0.5);
        Y = X + 2 + NOISE;
    output;
    end;

    CATE = "Varying Spread";
    do i = 0 to &split.;
        SUBJID = "07-"||put(i, z4.);
        TRT = "Y ~ X + N(0, 1.0)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(1);
        Y = X + NOISE;
    output;
    end;

    do i = 0 to &split.;
        SUBJID = "08-"||put(i, z4.);
        TRT = "Y ~ X + N(0, 2.0)";
        X = i * 1 / &split.;
        NOISE = rannor(&seed.)*sqrt(2);
        Y = X + NOISE;
    output;
    end;
run;

%GPCT(_DATSRC = dummy, _STRATUM = CATE TRT, _X = X, _Y = Y, _TREND = odds,
_ALPHA = 0.05);
```

The output GPCT dataset:

	CATE	TRT	Pc	Pd	GPCT	GPCT_SE	GPCT_LCI	GPCT_UCI	PVALUE
1	Varying Intercept	Y ~ X + 1 + N(0, 0.5)	0.6225662449	0.3764347541	1.653849009	0.0691921107	1.5236461658	1.7951783071	0
2	Varying Intercept	Y ~ X + 2 + N(0, 0.5)	0.6244903947	0.3745106043	1.6674838778	0.0674656866	1.5403605669	1.8050984573	0
3	Varying Slope	Y ~ 10X * (1 - X) + N(0, 0.5)	0.5140374111	0.4849635879	1.0599505283	0.0577956175	0.9525162072	1.1795023685	0.2856222217
4	Varying Slope	Y ~ 2X + N(0, 0.5)	0.7166439954	0.2823570036	2.5380776327	0.0992508907	2.350817302	2.7402546612	0
5	Varying Slope	Y ~ N(0, 0.5)	0.5138078705	0.4851931285	1.0589759834	0.0443133498	0.9755896417	1.1494895861	0.1708802578
6	Varying Slope	Y ~ X + N(0, 0.5)	0.6372768091	0.3617241899	1.7617754823	0.0709170686	1.6281221752	1.9064004516	0
7	Varying Spread	Y ~ X + N(0, 1.0)	0.5790453303	0.4199556687	1.3788248938	0.0592789111	1.267400774	1.5000449161	7.904788E-14
8	Varying Spread	Y ~ X + N(0, 2.0)	0.5736740782	0.4253269208	1.3487838415	0.0563436612	1.2427521859	1.4638621213	7.922552E-13

We've ensured that it produce the exact same result from the R package provided by author of the thesis.

Changing the intercept did not change the strength of association reported by GPCT, while changing either the slope of the function or the spread of Y values introduced by the random noise term impacted the results.
For monotonic relationships, increasing the slope of resulted in a larger GPCT effect size measure, while increasing the standard deviation of the random noise term resulted in a smaller GPCT effect size measure. This reflects the nature of GPCT as an extension of the GPC family of methods, which estimate the chance that a random treatment patient will

have a better outcome than a random control patient. In the two-group case, this chance may be increased by either increasing the strength of the effect (e.g. the difference in mean outcome score increases) or by reducing the variability of patient outcomes (e.g. individual outcomes become tightly clustered around a small difference in mean outcome scores). This same logic applies when the explanatory variable is continuous or has three or more ordered groups.

When the slope of TRT was zero (i.e. X has no impact on Y), there was no GPCT effect. The quadratic relationship also showed no GPCT effect, as the strong non-monotonicity of this relationship meant that in a random pair of observations, a higher value of X was not associated with a higher value of Y.

The output GPCT_OVERALL_TEST dataset:

	V	DF	PVALUE
1	324.76288732	7	0

For hypothesis testing H_0: $GPCT_1 = GPCT_2 = \cdots = GPCT_8$, a statistic $V \sim \chi^2$ with degrees of freedom 7 (8 stratum - 1) and its p-value given that H_0 is true are provided, and it comes into a conclusion that there is sufficient evidence to reject the null hypothesis.